

CHAPTER 1

InFocus

COMPLEX FORMULAS

Most spreadsheet users are comfortable with simple formulas such as $=B2+B3$ and simple functions such as **SUM**, but there is much more that can be done with formulas and functions. Simple formulas and functions are suitable for many uses, but there may be occasions when you need to create formulas of much greater complexity. Excel provides techniques for doing this and ensuring that the formulas are correct and accurate.

In this session you will:

- ✓ gain an understanding of how to scope a formula
- ✓ gain an understanding of how to develop a long-hand complex formula
- ✓ learn how to prepare a worksheet prior to building a complex formula
- ✓ learn how to commence a complex formula
- ✓ learn how to add more operations to a complex formula
- ✓ learn how to edit in a complex formula
- ✓ learn how to add more complexity to a long formula
- ✓ learn how to copy nested functions
- ✓ learn how to switch to manual recalculation
- ✓ learn how to paste values from formulas
- ✓ learn how to document a formula using comments.

SCOPING A FORMULA

Complex formulas are created by *embedding* one or more formulas and functions within another – this is sometimes referred to as **nesting**. For example, you may wish to sum half a dozen cells,

then multiply the result by 20% of another value. There are several components to this formula (for example, *summing* and *dividing*) that can be nested together to create a more complex formula.

The Scope of a Formula

All formulas perform an operation and have a specific outcome. Complex formulas perform several operations within the one formula but still have a specific outcome. In these formulas the several operations are *nested* within the one formula and act as building blocks to achieve the outcome. There are several recommended steps to *scoping* a complex formula:

1. Determine what the *outcome* of the formula should be
2. Determine the operations (and therefore the *sub-formulas* and *functions*) required to achieve this outcome
3. Translate these operations into Excel formula-jargon by writing them out long-hand
4. Commence by entering the *base operation* (either a formula or a function) and testing it
5. Add the next component and test the result
6. Repeat step 4 until the outcome is achieved.

The Case Study

Our case study spreadsheet calculates a *dividend* payable to superannuation investors. Here are the investors and their investment details:

We need to create a formula in column **G**, the *outcome* of which is to calculate the dividend payable to each client.

The dividend is calculated by multiplying the original investment (in column **E**) by a percentage based on the investment scale originally chosen by the investor (in column **F**).

The percentage to be used is based on a sliding scale which has been entered into a table as shown at the right. We can use a **VLOOKUP** function to extract the percentage from the table.

But there's a further complication: investments made earlier than June 30, 2004, are entitled to a 5% addition as a bonus to their dividend. Given that there are decisions to be made based on the sign-up date we can use the **IF** function to determine whether a bonus is payable or not.

	A	B	C	D	E	F	G	H
1	Alpheius Global Enterprises							
2	Superannuation Dividends							
3								
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend	
5	Pedro	Kavana	Mr	15/05/2005	10,000	C		
6	Jessica	Dunn	Miss	12/04/2004	12,000	A		
7	Tim	Nyguen	Mr	13/05/2004	3,000	A		
8	Fabian	Considine	Mr	12/01/2004	12,500	C		
9	Rose	Jovanovski	Ms	13/02/2004	4,500	C		
10	India	Beaumont	Ms	14/03/2005	2,300	B		
11	Bryn	Underwood	Mr	12/03/2005	1,200	A		
12	Sylvia	Schenk	Ms	1/03/2004	2,300	B		
13	Courtney	Perera	Miss	23/02/2004	15,000	C		
14	Shivanthe	Rasheed	Ms	21/01/2005	23,000	C		
15								
16								

	A	B	C	D	E
1					
2					
3	Bonus Eligibility Date:		30/06/2004		
4	Bonus:		5%		
5					
6					
7	Investment	Risk (A)	Growth (B)	Cons'tive (C)	
8	1,000	3%	2%	1%	
9	5,000	4%	3%	1%	
10	10,000	5%	4%	2%	
11	15,000	6%	5%	2%	
12	20,000	7%	6%	4%	
13	25,000	8%	8%	5%	
14					
15					

LONG-HAND FORMULAS

The best way to develop a **complex formula** is by developing each of the components first and then combining them. By writing each of the parts in sentence form, you will be able to understand

the logic of each more easily. You can then establish where the individual parts go in the overall scheme of your formula, create a base function, and then build your formula from there.

The Formula's Outcome

The formula's **outcome** can usually be expressed as a pseudo-formula. For example, our case study formula can be written as follows:

*=Original Investment Amount * (Dividend Percentage + Applicable Bonus Percentage)*

The next step is to add more detail to each of these components.

The Original Investment Amount

In our case study this will be the value that is in column **E**. So, our formula for the first client begins with:

=E5

The Dividend Percentage

We'll use a **VLOOKUP** function to find the appropriate percentage from the percentages table. The **VLOOKUP** function has three arguments – the value to look up, the table location, and the column to take the result from. Writing it out long-hand results in:

VLOOKUP(Original Investment, Percentages Table, Column based on Original Scale)

Writing this for our first client will look like:

VLOOKUP(E5,A8:D13...

We can't complete the formula yet because we haven't worked out how to choose the correct column from the table. Remember, this is based on the original scale chosen by the client when the investment was made. The logic for the column would be: if the scale is **A** then select column **2**; if the scale is **B** then select column **3**, if the scale is **C** then select column **4**. Therefore we need to use an **IF** function. **IF** functions only return either a true or a false answer. For example the first part of the formula would be as follows:

IF(Scale = A,2,3)

This formula will return the value in column **2** if the scale is **A**. If it is either **B** or **C** (or anything else) it will return the value in column **3**. To work around this we can nest another **IF** function to test to see if it is **B**:

IF(Scale = A, 2, IF(Scale = B, 3,4))

Here, a second **IF** statement has been used in the false position of the first **IF** statement. If the scale is anything but **A** the first **IF** function will default to the second (nested) **IF** function. If the scale is **B** the true position from the second **IF** statement will be returned. If the scale is neither **A** nor **B** the false position of the second **IF** function will be returned. Writing it for the first client the formula so far appears as:

=E5 * VLOOKUP(E5, A8:D13, IF(F5="A",2, IF(F5="B",3,4)))

Notice how the number of left and right brackets match.

The Bonus

The bonus is a percentage amount (as shown in cell **C4**) and is based on a cut-off date (as shown in **C3**). A bonus is paid if the investment was started before the date in cell **B4**. So an **IF** function should be able to provide the desired outcome:

IF(Start is earlier than bonus date, add bonus to the percentage, otherwise add nothing)

Translating this into a formula for the first client the function would be:

IF(D5<C3, C4, 0)

The Complete Formula

=E5*(VLOOKUP(E5,A8:D13,IF(F5="A",2,IF(F5="B",3,4)))+IF(D5<C3,C4,0))

Remember that, firstly, we are dealing with some values in a different worksheet, so the addressing will need to be different than that shown above. Secondly, we have enclosed the lookup table calculation and the calculation of the bonus together within brackets so that they are performed before the result is multiplied by the investment.

PREPARING FOR COMPLEX FORMULAS

Complex formulas are created by *nesting* formulas and functions within formulas. Since formulas and functions usually rely on cell referencing, complex formulas end up with many

cell and range addresses written into them. Excel allows you to give more meaningful **names** to cells and ranges in a workbook thus making it easier to work with and understand complex formulas.

Try This Yourself:

Open
File

Before starting this exercise you **MUST** open the file *E1318 Complex Formulas_1.xlsx*...

- 1 Click on the **Clients** worksheet tab
Spend a few moments studying the worksheet. This worksheet is where our complex formula will be built...
- 2 Click on the **Constants** worksheet tab
This worksheet contains key information (constants) used to calculate dividends for the clients...
- 3 Click in cell **A8**, hold down **Shift**, then click in cell **D13** to select the range **A8:D13**
This is the range that represents a lookup table to be used in the complex formula...
- 4 Click in the **Name** box to the left of the **Formula Bar**, as shown
- 5 Type **Dividend**, then press **Enter** to name the range **Dividend**, as shown
- 6 Click in cell **C3**, click in the **Name** box, type **BonusDate**, then press **Enter**
Range names can't contain spaces, so we've capitalised the first letter of each word and joined them together...
- 7 Click in cell **C4**, click in the **Name** box, type **BonusRate**, then press **Enter**

	A	B	C	D	E	F	G
1							
2							
3		Bonus Eligibility Date:	30/06/2013				
4		Bonus:	5%				
5							
6							
7	Investment	Risk (A)	Growth (B)	Const'tive (C)			
8	1,000	3%	2%	1%			
9	5,000	4%	3%	1%			
10	10,000	5%	4%	2%			
11	15,000	6%	5%	2%			
12	20,000	7%	6%	4%			
13	25,000	8%	8%	5%			
14							
15							

	A	B	C	D	E	F	G
1							
2							
3		Bonus Eligibility Date:	30/06/2013				
4		Bonus:	5%				
5							
6							
7	Investment	Risk (A)	Growth (B)	Const'tive (C)			
8	1,000	3%	2%	1%			
9	5,000	4%	3%	1%			
10	10,000	5%	4%	2%			
11	15,000	6%	5%	2%			
12	20,000	7%	6%	4%			
13	25,000	8%	8%	5%			
14							
15							

For Your Reference...

To **name** a **range** or **cell** in a **worksheet**:

1. Select the desired range or cell
2. Click in the **Name** box next to the **Formula Bar**
3. Type the desired name, then press **Enter**

Handy to Know...

- If you make a mistake with a range name you can edit and/or delete it (to start again) using the **Name Manager** in the **Defined Names** group on the **FORMULAS** tab.

CREATING THE BASE FORMULA

Complex formulas, like simple formulas, need to have a starting point. It is recommended that when working with a nested formula, you enter the most intricate of the formulas or functions in

the operation as a starting point, thus making it the base formula. In our case study the calculation of the dividend is arguably the most intricate.

Try This Yourself:

Same
File

Continue using the previous file with this exercise, or open the file E1318 Complex Formulas_2.xlsx...

- 1 Click on the **Clients** worksheet tab, then click in cell **G5**
- 2 Type = then click in cell **E5**
This is the investment amount...
- 3 Type ***VLOOKUP(**
- 4 Click in cell **E5** again, then type , (a comma)
- 5 Click on the **FORMULAS** tab, then click on **Use in Formula** in the **Defined Names** group to display a list of created names
- 6 Select **Dividend** to paste the name into the formula
- 7 Type **,2)** then press **Enter** to complete the formula
For testing purposes we'll use just the first column of the dividend table to make sure the formula works...
- 8 Examine the formula to see if it is producing the correct result (you may need a calculator)

E5								
	A	B	C	D	E	F	G	H
1	Alpheius Global Enterprises							
2	Superannuation Dividends							
3								
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend	
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	=E5	
6	Jessica	Dunn	Miss	12/04/2013	12,000	A		
7	Tim	Nyguen	Mr	13/05/2013	3,000	A		
8	Fabian	Considine	Mr	12/01/2013	12,500	C		
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C		
10	India	Beaumont	Ms	14/03/2013	2,300	B		

G5								
	A	B	C	D	E	F	G	H
1	Alpheius Global Enterprises							
2	Superannuation Dividends							
3								
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend	
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	=E5*VLOOKUP(E5,Dividend	
6	Jessica	Dunn	Miss	12/04/2013	12,000	A		
7	Tim	Nyguen	Mr	13/05/2013	3,000	A		
8	Fabian	Considine	Mr	12/01/2013	12,500	C		
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C		
10	India	Beaumont	Ms	14/03/2013	2,300	B		

G6								
	A	B	C	D	E	F	G	H
1	Alpheius Global Enterprises							
2	Superannuation Dividends							
3								
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend	
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	500	
6	Jessica	Dunn	Miss	12/04/2013	12,000	A		
7	Tim	Nyguen	Mr	13/05/2013	3,000	A		
8	Fabian	Considine	Mr	12/01/2013	12,500	C		
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C		
10	India	Beaumont	Ms	14/03/2013	2,300	B		

- 7 The formula at this point takes 10,000 (E5) to lookup the Dividend table. It will then extract the percentage from the second column (5%) and multiply this by the 10,000 to arrive at the 500 dividend.

For Your Reference...

To **start** a **complex formula**:

1. Determine which is the most intricate operation
2. Type the formula or function required for this operation
3. Test the results

Handy to Know...

- Range names make it much easier to reference ranges in formulas – you can simply select the range name from the list of names in the **Defined Names** group on the **FORMULAS** tab, rather than having to reference the parameters of the range.

ADDING MORE OPERATIONS

Once the base operation of a complex formula has been entered and tested you are ready to add more operations. Excel provides some tools to assist when adding more operations to a

formula. For example, you can add multiple lines to the formula to make it easier to read. Also, if you don't add the correct number of left and right brackets, Excel will attempt to correct this for you.

Try This Yourself:

Same File


Continue using the previous file with this exercise, or open the file E1318 Complex Formulas_3.xlsx...

- 1 Double click in cell **G5** to place it in edit mode
- 2 Click on the expand arrow at the right of the **Formula Bar** to expand the **Formula Bar**
- 3 In the **Formula Bar**, click to the left of **2**, press **Del** to remove it, then press **Alt** + **Enter** to start a new line
- 4 Type **IF(F5="A",2,3)**, then press **Enter**

We left a bracket out, but Excel is suggesting a correction. Formulas must have the same number of left and right brackets...

- 5 Click on **[Yes]** to accept Excel's correction, then click in cell **G5** to review the operation of the formula
Since cell **F5** doesn't equal **A**, the lookup table is taking the percentage from column 3. 4% of 10,000 is 400...
- 6 Click in cell **F5**, type **A** to change the scale, then press **Enter** to see the formula change to show **500**

SUM		:	X	✓	fx	=E5*VLOOKUP(E5,Dividend,			
	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	=E5*VLOOKUP(E5,Dividend,		
6	Jessica	Dunn	Miss	12/04/20	VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])				
7	Tim	Nyguen	Mr	13/05/2013	3,000	A			
8	Fabian	Considine	Mr	12/01/2013	12,500	C			
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C			
10	India	Beaumont	Ms	14/03/2013	2,300	B			
11	Bryn	Underwood	Mr	12/03/2013	1,200	A			
12	Sylvia	Schenk	Ms	1/03/2012	2,300	B			
13	Courtney	Perera	Miss	23/02/2012	15,000	C			



G5

:

X

✓

fx

=E5*VLOOKUP(E5,Dividend,IF(F5="A",2,3))

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	400		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A			
7	Tim	Nyguen	Mr	13/05/2013	3,000	A			
8	Fabian	Considine	Mr	12/01/2013	12,500	C			
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C			
10	India	Beaumont	Ms	14/03/2013	2,300	B			
11	Bryn	Underwood	Mr	12/03/2013	1,200	A			
12	Sylvia	Schenk	Ms	1/03/2012	2,300	B			
13	Courtney	Perera	Miss	23/02/2012	15,000	C			

For Your Reference...

To **add** more **operations** to a **complex formula**:

1. Place the formula in edit mode by double clicking on it or pressing **F2**
2. Make the desired changes, then press **Enter**

Handy to Know...

- You can choose to make changes to a formula either in the **Formula Bar** or in the cell that contains the formula. The method you choose doesn't make any difference to the formula.

EDITING A COMPLEX FORMULA

When you place a formula in edit mode by either double clicking on the cell containing the formula or by selecting the cell and pressing **F2**, the formula will appear colour-coded in the cell. The

colouring allows you to see which cells and ranges are referenced by the formula. It also shows you sets of left and right brackets so that you can see whether the brackets are balanced.

Try This Yourself:

Same
File

Continue using the previous file with this exercise, or open the file E1318 Complex Formulas_4.xlsx...

1 Double click in cell **G5** to place it in edit mode
Notice the colouring used for the matching brackets...

2 Click anywhere in the **IF** part of the formula, then click on **[value-if-false]**, as shown, to see that part of the function

We need to extend our **IF** function to include scale A, B, and C...

3 Type **IF(F5="B",3,4)**
Notice how the colours of the brackets are updated...

4 Press **Enter** to complete the formula changes

5 Click in cell **F5**, type **B**, then press **Ctrl** + **Enter**
The dividend should change to 400 (10,000 x 4%)...

6 Type **C**, then press **Ctrl** + **Enter**
The dividend should change to 200 (10,000 x 2%)

SUM : X ✓ fx =E5*VLOOKUP(E5,Dividend, IF(F5="A",2,3))

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	A	=E5*VLOOKUP(E5,Dividend, IF(F5="A",2,3))		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A			
7	Tim	Nyguen	Mr	13/05/2013	3,000				
8	Fabian	Considine	Mr	12/01/2013	12,500	C			
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C			

2

SUM : X ✓ fx =E5*VLOOKUP(E5,Dividend, IF(F5="A",2,IF(F5="B",3,4)))

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	A	=E5*VLOOKUP(E5,Dividend, IF(F5="A",2,IF(F5="B",3,4)))		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A			
7	Tim	Nyguen	Mr	13/05/2013	3,000				
8	Fabian	Considine	Mr	12/01/2013	12,500	C			
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C			

3

F5 : X ✓ fx C

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	200		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A			
7	Tim	Nyguen	Mr	13/05/2013	3,000	A			
8	Fabian	Considine	Mr	12/01/2013	12,500	C			
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C			

6

For Your Reference...

To **edit** a **complex formula**:

1. Double click in the cell containing the formula, or click in the cell and press **F2**
2. Click on the relevant function to change, then click on the hyperlinked text in the tooltip
3. Make the appropriate changes

Handy to Know...

- Clicking on the hyperlinked tooltip text when you have placed the cursor on a function within an edited formula makes it easier to select parts of the function to change.
- The outer brackets of a complex formula are always black.

ADDING MORE COMPLEXITY

As you add more operations to a formula you are adding to the complexity of that formula. Make sure you remember the **BODMAS** rules. Even where operations nested in the formula adhere to

the rules of **BODMAS** it is often helpful to enclose specific operations within their own brackets simply to keep them understandable and readable.

Try This Yourself:

Same File

Continue using the previous file with this exercise, or open the file *E1318 Complex Formulas_5.xlsx*...

- 1 Double click in cell **G5** to place it in edit mode
- 2 Click to the left of **VLOOKUP**, then type (

Notice that the colour of the brackets no longer matches. You have a black left bracket but no matching right one...
- 3 Press **Ctrl** + **End** to move to the end of the formula, then press **Alt** + **Enter** to start a new line
- 4 Type **+IF(D5<=**

This is the start of the operation that will calculate whether a bonus is due in addition to the dividend...
- 5 Click on the **FORMULAS** tab, click on **Use in Formula** in the **Defined Names** group, then click on **BonusDate**
- 6 Type **,** (a comma), click on **Use in Formula** again, then click on **BonusRate**
- 7 Type **,0)** to complete the **IF** function, then **)** to add the remaining bracket
- 8 Press **Ctrl** + **Enter** to complete the formula

SUM

:

✕

✓

fx

=E5*(VLOOKUP(E5,Dividend, IF(F5="A",2,IF(F5="B",3,4)))

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	=E5*(VLOOKUP(E5,Dividend, IF(F5="A",2,IF(F5="B",3,4)))		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A			
7	Tim	Nyguen	Mr	13/05/2013	3,000	A			
8	Fabian	Considine	Mr	12/01/2013	12,500	C			
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C			

2

SUM		✕ ✓ fx		=E5*(VLOOKUP(E5,Dividend, IF(F5="A",2,IF(F5="B",3,4))) +IF(D5<=					
	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	=E5*(VLOOKUP(E5,Dividend, IF(F5="A",2,IF(F5="B",3,4)))		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A			
7	Tim	Nyguen	Mr	13/05/2013	3,000	A			
8	Fabian	Considine	Mr	12/01/2013	12,500	C			
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C			

4

G5		✕ ✓ fx		=E5*(VLOOKUP(E5,Dividend, IF(F5="A",2,IF(F5="B",3,4))) +IF(D5<=BonusDate,BonusRate,0))					
	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	700		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A			
7	Tim	Nyguen	Mr	13/05/2013	3,000	A			
8	Fabian	Considine	Mr	12/01/2013	12,500	C			
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C			

For Your Reference...

To **add** more **complexity** to a **complex formula**:

1. Double click in the cell containing the formula, or click in the cell and press **F2**
2. Remember to adhere to the rules of **BODMAS** and add brackets if required

Handy to Know...

- Placing new lines in a formula in the **Formula Bar** can make the formula a little easier to understand. Where possible place each operation of the complex formula on a new line.

COPYING NESTED FUNCTIONS

Formulas that contain nested operations can be copied in a worksheet or workbook just like any other formula. However, you should be especially careful of the cell addresses used in the formula

to ensure that they adjust as required. The need for absolute cell addressing can sometimes be difficult to identify in longer and complex formulas.

Try This Yourself:

Same
File

Continue using the previous file with this exercise, or open the file *E1318 Complex Formulas_6.xlsx*...

- 1 Double click in cell **G5** to place it in edit mode
The references to cells E5, F5, and D5 all need to adjust as the formula is copied down. The named ranges will need to lock into their specific address – fortunately named ranges are absolute as a default...
- 2 Press **Esc** to cancel edit mode as there is nothing we need to do to the addressing
- 3 Double-click on the fill handle of cell **G5** to fill the formula down
- 4 Click in cell **G15**, click on the **FORMULAS** tab, then click on **AutoSum** in the **Function Library** group to commence a **Sum** function
- 5 Press **Enter** to complete the function
- 6 Click in cell **G15** again, click on the **HOME** tab, click on **Cell Styles** in the **Styles** group, then click on **Total**
- 7 Click in cell **A1** to see the formatting

3

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	700		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A	1,200		
7	Tim	Nyguen	Mr	13/05/2013	3,000	A	240		
8	Fabian	Considine	Mr	12/01/2013	12,500	C	875		
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C	270		
10	India	Beaumont	Ms	14/03/2013	2,300	B	161		
11	Bryn	Underwood	Mr	12/03/2013	1,200	A	96		
12	Sylvia	Schenk	Ms	1/03/2012	2,300	B	161		
13	Courtney	Perera	Miss	23/02/2012	15,000	C	1,050		
14	Shivanthe	Rasheed	Ms	21/01/2013	23,000	C	2,070		
15									
16									
17									

5

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	700		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A	1,200		
7	Tim	Nyguen	Mr	13/05/2013	3,000	A	240		
8	Fabian	Considine	Mr	12/01/2013	12,500	C	875		
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C	270		
10	India	Beaumont	Ms	14/03/2013	2,300	B	161		
11	Bryn	Underwood	Mr	12/03/2013	1,200	A	96		
12	Sylvia	Schenk	Ms	1/03/2012	2,300	B	161		
13	Courtney	Perera	Miss	23/02/2012	15,000	C	1,050		
14	Shivanthe	Rasheed	Ms	21/01/2013	23,000	C	2,070		
15							=SUM(G5:G14)		
16							SUM(number1, [number2], ...)		
17									

7

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	700		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A	1,200		
7	Tim	Nyguen	Mr	13/05/2013	3,000	A	240		
8	Fabian	Considine	Mr	12/01/2013	12,500	C	875		
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C	270		
10	India	Beaumont	Ms	14/03/2013	2,300	B	161		
11	Bryn	Underwood	Mr	12/03/2013	1,200	A	96		
12	Sylvia	Schenk	Ms	1/03/2012	2,300	B	161		
13	Courtney	Perera	Miss	23/02/2012	15,000	C	1,050		
14	Shivanthe	Rasheed	Ms	21/01/2013	23,000	C	2,070		
15							6,823		
16									
17									

For Your Reference...

To **copy** a **nested function**:

1. Check the cell references and adjust for absolute addressing if required
2. Copy the formula using your preferred copying methodology

Handy to Know...

- Remember, a **named range** refers to a specific and absolute location in a workbook. Therefore there is no need to mark range names as absolute in formulas.

SWITCHING TO MANUAL RECALCULATION

Recalculation refers to processing the formulas in a spreadsheet to calculate new results. Formulas are usually recalculated each time a value in a dependent cell changes, but you can

turn off **automatic** recalculation and instead set Excel to **manual**. This means that no formulas will be recalculated unless you specifically request Excel to perform the calculations.

Try This Yourself:


Same
File

Continue using the previous file with this exercise, or open the file E1318 Complex Formulas_7.xlsx...

- 1 Click in cell **F5**, type **A**, then press **Enter**
The dividend and total will update to reflect the change...
- 2 Click on the **FORMULAS** tab, then click on **Calculation Options** in the **Calculation** group
- 3 Select **Manual**
- 4 Click in cell **F5**, type **C**, then press **Enter**
This time the dividend and total won't change...
- 5 Click on **Calculate Now** in the **Calculation** group on the **FORMULAS** tab to force a manual update
- 6 Click on **Calculation Options** in the **Calculation** group on the **FORMULAS** tab, then click on **Automatic** to restore the settings to automatic calculation
- 7 Click on the collapse arrow at the right of the **Formula Bar** to collapse the **Formula Bar** back to its default size

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	A	1,000		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A	1,200		
7	Tim	Nyguen	Mr	13/05/2013	3,000	A	240		
8	Fabian	Considine	Mr	12/01/2013	12,500	C	875		
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C	270		
10	India	Beaumont	Ms	14/03/2013	2,300	B	161		
11	Bryn	Underwood	Mr	12/03/2013	1,200	A	96		
12	Sylvia	Schenk	Ms	1/03/2012	2,300	B	161		
13	Courtney	Perera	Miss	23/02/2012	15,000	C	1,050		
14	Shivanthe	Rasheed	Ms	21/01/2013	23,000	C	2,070		
15							7,123		
16									
17									

1



2

	A	B	C	D	E	F	G	H	I
1	Alpheius Global Enterprises								
2	Superannuation Dividends								
3									
4	Firstname	Lastname	Title	Joined	Investment	Scale	Dividend		
5	Pedro	Kavana	Mr	15/05/2012	10,000	C	1,000		
6	Jessica	Dunn	Miss	12/04/2013	12,000	A	1,200		
7	Tim	Nyguen	Mr	13/05/2013	3,000	A	240		
8	Fabian	Considine	Mr	12/01/2013	12,500	C	875		
9	Rose	Jovanovski	Ms	13/02/2012	4,500	C	270		
10	India	Beaumont	Ms	14/03/2013	2,300	B	161		
11	Bryn	Underwood	Mr	12/03/2013	1,200	A	96		
12	Sylvia	Schenk	Ms	1/03/2012	2,300	B	161		
13	Courtney	Perera	Miss	23/02/2012	15,000	C	1,050		
14	Shivanthe	Rasheed	Ms	21/01/2013	23,000	C	2,070		
15							7,123		
16									
17									

4

For Your Reference...

To **turn off automatic formula calculation**:

1. Click on the **FORMULAS** tab
2. Click on **Calculation Options** in the **Calculation** group
3. Select **Manual**

Handy to Know...

- The **Calculation** settings are global and will affect every spreadsheet you work with. You can't force only one workbook to require manual recalculation without affecting others without doing some complex programming.

PASTING VALUES FROM FORMULAS

Sometimes it's useful to be able to take the results of a calculation and use the value elsewhere in a spreadsheet without keeping the formula. For example, you may want to keep a

copy of some data at a certain point in time, knowing that it won't change at a later date. You can do this by copying formulas and pasting only the values.


Try This Yourself:

Same File

Continue using the previous file, or open the file E1318 Complex Formulas_8.xlsx...

- 1 Select the range **B4:B14**, hold down **Ctrl**, then select the range **G4:G14**
The first range contains text while the second contains formulas (values)...
- 2 Click on the **HOME** tab, then click on **Copy** in the **Clipboard** group to copy the ranges
- 3 Click in cell **J4**
This will be the start of where we'll paste the copied data...
- 4 Click on the bottom half of **Paste** in the **Clipboard** group to see the paste options
- 5 Point to the various options to see a Live Preview of how the data will appear when pasted
- 6 Click on the third option in the **Paste Values** section to paste the data as values

	A	B	C	D	E	F	G	H	I	J	K
1		Alpheius Global Enterprises									
2		Superannuation Dividends									
3											
4		Firstname	Lastname	Title	Joined	Investment	Scale			Dividend	
5		Pedro	Kavana	Mr	15/05/2012	10,000	C			700	
6		Jessica	Dunn	Miss	12/04/2013	12,000	A			1,200	
7		Tim	Nyguen	Mr	13/05/2013	3,000	A			240	
8		Fabian	Considine	Mr	12/01/2013	12,500	C			875	
9		Rose	Jovanovski	Ms	13/02/2012	4,500	C			270	
10		India	Beaumont	Ms	14/03/2013	2,300	B			161	
11		Bryn	Underwood	Mr	12/03/2013	1,200	A			96	
12		Sylvia	Schenk	Ms	1/03/2012	2,300	B			161	
13		Courtney	Perera	Miss	23/02/2012	15,000	C			1,050	
14		Shivanthe	Rasheed	Ms	21/01/2013	23,000	C			2,070	
15										6,823	
16											



	A	B	C	D	E	F	G	H	I	J	K
1		Alpheius Global Enterprises									
2		Superannuation Dividends									
3											
4		Firstname	Lastname	Title	Joined	Investment	Scale			Dividend	
5		Pedro	Kavana	Mr	15/05/2012	10,000	C			700	
6		Jessica	Dunn	Miss	12/04/2013	12,000	A			1,200	
7		Tim	Nyguen	Mr	13/05/2013	3,000	A			240	
8		Fabian	Considine	Mr	12/01/2013	12,500	C			875	
9		Rose	Jovanovski	Ms	13/02/2012	4,500	C			270	
10		India	Beaumont	Ms	14/03/2013	2,300	B			161	
11		Bryn	Underwood	Mr	12/03/2013	1,200	A			96	
12		Sylvia	Schenk	Ms	1/03/2012	2,300	B			161	
13		Courtney	Perera	Miss	23/02/2012	15,000	C			1,050	
14		Shivanthe	Rasheed	Ms	21/01/2013	23,000	C			2,070	
15										6,823	
16											

For Your Reference...

To **paste values** from **formulas**:

1. Select the data to copy, click on the **HOME** tab, then click on **Copy** in the **Clipboard** group
2. Click where you want to paste the data, then click on the bottom half of **Paste** in the **Clipboard** group and select an option

Handy to Know...

- When you copy formulas, you have the option to paste formulas, values and links. A link is a reference to the cell containing the formula. For example, if the cell containing the copied formula is cell **G12**, the link created by pasting will be **=G\$12**.

DOCUMENTING FORMULAS

Complex formulas can be difficult enough to decipher just after they've been written, let alone after a week or a month. Therefore, as soon as you have completed a complex formula and it is

working, it is a good idea to document it. There are many options for doing this but one simple way is to place a comment in the cell of the formula.

Try This Yourself:

Same
File

Continue using the previous file with this exercise, or open the file *E1318 Complex Formulas_9.xlsx...*

- 1 Click on the expand arrow for the **Formula Bar**, then click in cell **G5** to see the first instance of the formula
- 2 Click on the **REVIEW** tab, then click on **New Comment** in the **Comments** group to insert a **Comment** box
- 3 Click under the name, then type the following text:

A dividend percentage is calculated by using the investment to look up the appropriate scale from the Dividend table (see Constants worksheet). A bonus is added if the date joined is on or earlier than 30/6/2004.

The investment amount is then multiplied by the calculated dividend percentage.

- 4 Resize the **Comment** box to fit the text so it appears as shown
- 5 Click in cell **F5** to deselect the commented cell
- 6 Point to the red marker in cell **G5** to see the **Comment** box
- 7 Click on the collapse arrow at the right of the **Formula Bar** to collapse the **Formula Bar** back to one line again

Firstname	Lastname	Title	Joined	Investment	Scale	Dividend
Pedro	Kavana	Mr	15/05/2012	10,000	C	700
Jessica	Dunn	Miss	12/04/2013	12,000	A	1,200
Tim	Nyguen	Mr	13/05/2013	3,000	A	240
Fabian	Considine	Mr	12/01/2013	12,500	C	875
Rose	Jovanovski	Ms	13/02/2012	4,500	C	270
India	Beaumont	Ms	14/03/2013	2,300	B	161
Bryn	Underwood	Mr	12/03/2013	1,200	A	96
Sylvia	Schenk	Ms	1/03/2012	2,300	B	161
Courtney	Perera	Miss	23/02/2012	15,000	C	1,050
Shivanthe	Rasheed	Ms	21/01/2013	23,000	C	2,070
						6,823

Firstname	Lastname	Title	Joined	Investment	Scale	Dividend
Pedro	Kavana	Mr	15/05/2012	10,000	C	700
Jessica	Dunn	Miss	12/04/2013	12,000	A	1,200
Tim	Nyguen	Mr	13/05/2013	3,000	A	240
Fabian	Considine	Mr	12/01/2013	12,500	C	875
Rose	Jovanovski	Ms	13/02/2012	4,500	C	270
India	Beaumont	Ms	14/03/2013	2,300	B	161
Bryn	Underwood	Mr	12/03/2013	1,200	A	96
Sylvia	Schenk	Ms	1/03/2012	2,300	B	161
Courtney	Perera	Miss	23/02/2012	15,000	C	1,050
Shivanthe	Rasheed	Ms	21/01/2013	23,000	C	2,070
						6,823

For Your Reference...

To **document** a **formula** using a **comment**:

1. Click in the cell containing the formula
2. Click on the **REVIEW** tab, then click on **New Comment** in the **Comments** group
3. Type the comment text

Handy to Know...

- There is no need to place a comment in every cell that uses the formula – the worksheet would become too cluttered if you did. If the formula is filled as ours has been here then you really only need a comment in the first formula cell.